

V článku [Tabulka aneb Seznam čili ListObject](#) jsme se věnovali Tabulkám z pohledu práce na listu. Dnes si je osaháme prostřednictvím kódu VBA, kde jim není vyhrazen podle očekávání objekt Table, ale ListObject. Následující procedury najdete v příloženém sešitu a doporučuji je odkrokovat při současném zobrazení listu s Tabulkou.

	OBCHODNIK	MNOZSTVI	JEDNOTKOVA CENA	CENA
3	Sochor	10	100 Kč	1 000 Kč
4	Richtr	20	200 Kč	4 000 Kč
5	Žůrek	30	300 Kč	9 000 Kč
6	Kučera	40	400 Kč	16 000 Kč
7	Chvátalová	50	500 Kč	25 000 Kč
8	Průša	60	600 Kč	36 000 Kč
9	Špičková	70	700 Kč	49 000 Kč
10	Morávek	80	800 Kč	64 000 Kč
11	Koch	90	900 Kč	81 000 Kč
12	Heřmánek	100	1 000 Kč	100 000 Kč
20	Mach	110	1100	
21	Šebestová	120	1200	
22	Pažout	130	1300	
23	Kadrnožková	140	1400	

Tabulka pro VBA

V kódu najdete hodně poznámek ohledně posunu buněk pod Tabulkou. K němu (ne vždy) dochází, pokud v Tabulce mažete/přidáváte řádky nebo měníte zobrazení Řádku souhrnů. Pokud toto řešit nechcete, tak se držte toho, co bylo řečeno posledně – do prostoru pod Tabulkou již nekládejte žádné s ní nesouvisející hodnoty.

```
1 Sub TabulkaCasti()  
2  
3 Dim wshList As Worksheet  
4 Dim loTabulka As ListObject  
5  
6 Set wshList = Worksheets("Tabulka a VBA")  
7 Set loTabulka = wshList.ListObjects("MojeTabulka")  
8  
9 With loTabulka  
10  
11 'celá tabulka  
12 .Range.Select  
13  
14 'hlavička tabulky  
15 .HeaderRowRange.Select  
16  
17 'datová část tabulky  
18 .DataBodyRange.Select  
19  
20 'datová část tabulky (konstanty, tj. bez vzorců)  
21 .DataBodyRange.SpecialCells(xlCellTypeConstants).Select  
22  
23 'druhý sloupec tabulky  
24 .ListColumns(2).Range.Select  
25  
26 'datová část druhého sloupce tabulky  
27 .ListColumns(2).DataBodyRange.Select  
28  
29 'třetí datový řádek tabulky  
30 .ListRows(3).Range.Select  
31  
32 'řádek souhrnů (pokud je zobrazen)  
33 '.ShowTotals = True  
34 .TotalsRowRange.Select  
35  
36 End With  
37  
38 End Sub
```

```
1 Sub TabulkaVlastnosti()  
2  
3 Dim wshList As Worksheet  
4 Dim loTabulka As ListObject  
5  
6 Set wshList = Worksheets("Tabulka a VBA")  
7 Set loTabulka = wshList.ListObjects("MojeTabulka")  
8  
9 With loTabulka  
10  
11 'počet řádků včetně hlavičky (a řádku souhrnů)  
12 intPocetRadku = .Range.Rows.Count  
13  
14 'počet datových řádků  
15 intPocetZaznamu = .ListRows.Count  
16  
17 End With  
18  
19 End Sub
```

```
1 Sub TestBunkaTabulky()  
2  
3 Dim rngBunka As Range  
4 Dim blnBunkaTabulky As Boolean  
5  
6 Set rngBunka = ActiveCell  
7  
8 On Error Resume Next  
9  
10 'je buňka součástí Tabulky?  
11 blnBunkaTabulky = (rngBunka.ListObject.Name <> "")  
12  
13 On Error GoTo 0  
14  
15 If blnBunkaTabulky = True Then  
16  
17 'výběr datové oblasti Tabulky, jíž je buňka součástí  
18 rngBunka.ListObject.DataBodyRange.Select  
19  
20 End If  
21  
22 End Sub
```

```
1 Sub PridaniJednohoRadku()
2
3 'zobrazení řádku souhrnů není na překážku
4
5 Dim wshList As Worksheet
6 Dim loTabulka As ListObject
7
8 Set wshList = Worksheets("Tabulka a VBA")
9 Set loTabulka = wshList.ListObjects("MojeTabulka")
10
11 With loTabulka
12
13 'vložení nového řádku na konec tabulky
14 .ListRows.Add
15
16 'vložení nového řádku na konec tabulky
17 'buňky pod tabulkou se neposunou
18 .ListRows.Add AlwaysInsert:=False
19 'pozn. Zobrazení/skrytí řádku souhrnů posouvá
20 'buňky ležící pod Tabulkou vždy...
21
22 'vložení nového řádku nad 3. záznam
23 'tj. definujeme pozici nového řádku
24 .ListRows.Add (3)
25
26 End With
27
28 End Sub
```

```

1 Sub PridaniDatNaKonec()
2
3     'Zack Barresse
4
5     Dim wshList As Worksheet
6     Dim loTabulka As ListObject
7     Dim arrPridavanaData As Variant
8     Dim inPocetRadku As Integer
9     Dim intPocetSloupcu As Integer
10
11     Set wshList = Worksheets("Tabulka a VBA")
12     Set loTabulka = wshList.ListObjects("MojeTabulka")
13
14     'převzetí dat, zde z oblasti buněk do pole
15     arrPridavanaData = wshList.Range("B20:D23").Value
16
17     'rozměry datového bloku (pole)
18     inPocetRadku = UBound(arrPridavanaData, 1)
19     intPocetSloupcu = UBound(arrPridavanaData, 2)
20
21     'nutné skrytí řádku souhrnů, pokud je zobrazen
22     'pozor, dojde k posunu buněk ležících pod Tabulkou
23     loTabulka.ShowTotals = False
24
25     With loTabulka.DataBodyRange
26         'přidání dat na konec Tabulky
27         'neposouvá buňky ležící pod Tabulkou
28         .Resize(inPocetRadku, intPocetSloupcu).Offset(.Rows.Count).Value = _
29             arrPridavanaData
30     End With
31
32     'zobrazení řádku souhrnů
33     'pozor, dojde k posunu buněk ležících pod Tabulkou
34     loTabulka.ShowTotals = True
35
36     'oživení filtru
37     If loTabulka.ShowAutoFilter = True Then
38         loTabulka.AutoFilter.ApplyFilter
39     End If
40
41 End Sub

```

Pozn. Řádek souhrnů je sice chytrý, ale někdy nám prostě překáží.

```

1 Sub VlozeniDat()
2
3     'Zack Barresse
4
5     Dim wshList As Worksheet
6     Dim loTabulka As ListObject
7     Dim arrPridavanaData As Variant
8     Dim inPocetRadku As Integer
9     Dim intPocetSloupcu As Integer
10
11     Set wshList = Worksheets("Tabulka a VBA")
12     Set loTabulka = wshList.ListObjects("MojeTabulka")
13
14     'převzetí dat, zde z oblasti buněk do pole
15     arrPridavanaData = wshList.Range("B20:D23").Value
16
17     'rozměry datového bloku (pole)
18     inPocetRadku = UBound(arrPridavanaData, 1)
19     intPocetSloupcu = UBound(arrPridavanaData, 2)
20
21     With loTabulka.DataBodyRange
22         'přidání dat mezi 5 a 6. řádek Tabulky
23         'tj. pod 5. řádek
24         'neposouvá buňky ležící pod Tabulkou
25         .Resize(inPocetRadku).Offset(5).Insert Shift:=xlShiftDown
26         .Resize(inPocetRadku, intPocetSloupcu).Offset(5).Value = _
27             arrPridavanaData
28     End With
29
30     'oživení filtru
31     If loTabulka.ShowAutoFilter = True Then
32         loTabulka.AutoFilter.ApplyFilter
33     End If
34
35 End Sub

```

Pokud se nepletu, Tabulku průvodci na listu nativně nabízejí pro zdroje dat z Microsoft Access, Microsoft Excel a pravděpodobně i z Microsoft SQL nebo třeba MySQL. V případě textových souborů (CSV) to trochu skřípe. Je otázka, zda-li jít cestou QueryTable či OLEDB (ADO, schema.ini). O tom ale zase někdy jindy. Nyní se podíváme na filtrování.

```

1 Sub TabulkaFiltr()
2
3 Dim wshList As Worksheet
4 Dim loTabulka As ListObject
5
6 Set wshList = Worksheets("Tabulka a VBA")
7 Set loTabulka = wshList.ListObjects("MojeTabulka")
8
9 'zapnutí/vypnutí filtru Tabulky
10 loTabulka.Range.AutoFilter
11
12 'striktní obsah (Richtr)
13 loTabulka.Range.AutoFilter Field:=1, Criteria1:="=Richtr"
14
15 'striktní obsah (vše kromě položky Richtr)
16 loTabulka.Range.AutoFilter Field:=1, Criteria1:="<>Richtr"
17
18 'obsah (položky začínající na písmeno K)
19 '? ... jeden znak, * ... žádný nebo více znaků
20 loTabulka.Range.AutoFilter Field:=1, Criteria1:="=K*"
21
22 'striktní obsah (Koch nebo Morávek)
23 loTabulka.Range.AutoFilter Field:=1, Criteria1:="=Koch", Operator:=xlOr, _
24 Criteria2:="=Morávek"
25
26 'striktní obsah (Koch, Morávek, Sochor)
27 loTabulka.Range.AutoFilter Field:=1, Criteria1:=Array("Koch", "Morávek", _
28 "Sochor"), Operator:=xlFilterValues
29 'není nástroj pro "vše kromě položek z následujícího výčtu"
30 'lze řešit negací úlohy nebo přes rozšířený filtr, tzn.
31 'OBCHODNIK OBCHODNIK OBCHODNIK
32 '<>Koch <>Morávek <>Sochor
33
34 'v případě datumu je nutný americký formát mm/dd/yyyy
35 'a to bez ohledu na výstup ze Záznamníku maker
36 'tj. např. Criteria1:=">=02/23/2015"
37 'lze si pomoci funkcemi CLng, CDate
38
39 'oživení filtru (po aktualizaci dat)
40 'If loTabulka.ShowAutoFilter = True Then
41 ' loTabulka.AutoFilter.ApplyFilter
42 'End If
43
44 End Sub

```

Pozor na jednu nebezpečnou věc. V rámci filtru a parametru Criteria očekává Excel String a to i v případě čísel a testování rovnosti (regulérně bychom měli doplňovat znak „=“, např. Criteria1:="=6").

Běžně si Excel s předaným číselným typem (Integer) poradí i bez uvedeného rovnítka, ale rozhodně tento stav nesmí nastat při užití Operator:= xlFilterValues. Jestliže takto definovanému filtru předhodíte jednu hodnotu číselného typu, tak Excel začne dělat psí kusy, přestane fungovat překreslování a aplikace nepůjde korektně zavřít. Jedná se podle mě o neošetřený stav a bug.

```
1 Sub ResetFiltruTabulky()  
2  
3 Dim wshList As Worksheet  
4 Dim loTabulka As ListObject  
5  
6 Set wshList = Worksheets("Tabulka a VBA")  
7 Set loTabulka = wshList.ListObjects("MojeTabulka")  
8  
9 'pokud Tabulka používá filtr...  
10 If loTabulka.ShowAutoFilter Then  
11  
12 'pokud je filtr aktivní  
13 If loTabulka.AutoFilter.FilterMode Then  
14  
15 'zobrazení všech dat  
16 loTabulka.AutoFilter.ShowAllData  
17  
18 End If  
19  
20 End If  
21  
22 End Sub
```

Pucování tabulky od dat lze řešit dvojím způsobem. Osobně jsem si oblíbil druhý z nich.



```

1 Sub VycisteníTabulky1()
2
3 Dim wshList As Worksheet
4 Dim loTabulka As ListObject
5
6 Set wshList = Worksheets("Tabulka a VBA")
7 Set loTabulka = wshList.ListObjects("MojeTabulka")
8
9 'vymazání všech datových řádků tabulky
10 'nelze aplikovat na Tabulce bez datových řádků
11 If loTabulka.ListRows.Count > 0 Then
12
13     loTabulka.DataBodyRange.Delete
14     'výsledek:
15     'podobně DataBodyRange = Nothing
16     'Tabulka zobrazuje jeden fiktivní datový řádek (bez vzorců)
17     'Rows.Count = 2 pro hlavičku + fiktivní řádek
18     'ale (!) ListRows.Count = 0
19     'dojde k posunu buněk ležících pod Tabulkou
20
21     'pro další práci je nutné přidat alespoň jeden řádek
22     loTabulka.ListRows.Add
23     'dojde k navrácení případných vzorců ve sloupcích
24     'pozn. vzorce se uchovávají v ../xl/tables/table?.xml
25     'buňky ležící pod Tabulkou se neposunou
26
27 End If
28
29 End Sub

```

```

1 Sub VycisteniTabelky2()
2
3 Dim wshList As Worksheet
4 Dim loTabelka As ListObject
5
6 Set wshList = Worksheets("Tabelka a VBA")
7 Set loTabelka = wshList.ListObjects("MojeTabelka")
8
9 'operace proběhnou jen na minimálně jednořádkové
10 'vyplněné datové oblasti
11 On Error Resume Next
12
13 Application.ScreenUpdating = False
14
15 With loTabelka.DataBodyRange
16     'odstranění všech řádků kromě prvního
17     .Resize(.Rows.Count - 1).Offset(1).Delete
18     'vyčištění zbylého řádku od hodnot (vzorce ponechány)
19     .SpecialCells(xlCellTypeConstants).ClearContents
20 End With
21
22 On Error GoTo 0
23
24 Application.ScreenUpdating = True
25
26 End Sub

```

Příloha (váže se i k předchozímu článku):

[tabelka.zip](#)